Key Homework 25

Math 277, Fall 2005

Ordinary Differential Equations

Dr. Goutziers

Initializations

```
> restart;
  with(inttrans):
  with(student):
```

21: Page 375

```
> F:=s->(6*s^2-13*s+2)/(s*(s-1)*(s-6));
```

$$F := s \to \frac{6\,s^2 - 13\,s + 2}{s\,(s-1)\,(s-6)}$$

To find the inverse Laplace transform of this function we perform a partial fraction decomposition.

```
> e1:=convert(F(s), parfrac, s);
```

$$e1 := \frac{14}{3\,(s-6)} + \frac{1}{3\,s} + \frac{1}{s-1}$$

This means that

$$\mathrm{f}(t) = \frac{14\,\mathbf{e}^{(6\,t)}}{3} + \frac{1}{3} + \mathbf{e}^t$$

We verify this result by using Maple's **invlaplace** routine.

```
> e2:=invlaplace(F(s), s, t);
```

$$e2 := \frac{1}{3} + \frac{14}{3}\,\mathbf{e}^{(6\,t)} + \mathbf{e}^t$$

```
>
```

26: Page 375

```
> F:=s->(7*s^3-2*s^2-3*s+6)/(s^3*(s-2));
```

$$F := s \to \frac{7\,s^3 - 2\,s^2 - 3\,s + 6}{s^3\,(s-2)}$$

To find the inverse Laplace transform of this function we perform a partial fraction decomposition.

```
> e1:=convert(F(s), parfrac, s);
```

$$e1 := \frac{6}{s-2} - \frac{3}{s^3} + \frac{1}{s}$$

This means that

$$\mathrm{f}(t) = 6\,\mathbf{e}^{(2\,t)} - \frac{3\,t^2}{2} + 1$$

We verify this result by using Maple's **invlaplace** routine.

```
> e2:=invlaplace(F(s), s, t);
```

$$e2 := 1 + 6\,\mathbf{e}^{(2\,t)} - \frac{3\,t^2}{2}$$

```
>
```

29: Page 375

Solve the equation

$$s\,F(s) + 2\,F(s) = \frac{10\,s^2 + 12\,s + 14}{s^2 - 2\,s + 2}$$

for $F(s)$. Then perform a partial fraction decomposition.

First reset the variable *F*, because it was assigned in the previous problem.

```
> F:='F';
```

$$F := F$$

```
> e1:=s*F(s)+2*F(s)=(10*s^2+12*s+14)/(s^2-2*s+2);
```

$$e1 := s\,F(s) + 2\,F(s) = \frac{10\,s^2 + 12\,s + 14}{s^2 - 2\,s + 2}$$

```
> e2:=solve(e1, F(s));
```

$$e2 := \frac{2\,(5\,s^2 + 6\,s + 7)}{s^3 - 2\,s + 4}$$

To find the inverse Laplace transform of this function we perform a partial fraction decomposition.

```
> e3:=convert(e2, parfrac, s);
```

$$e3 := \frac{7\,s + 4}{s^2 - 2\,s + 2} + \frac{3}{s + 2}$$

Complete the square on the variable *s*.

```
> e4:=completesquare(e3, s);
```

$$e4 := \frac{7\,s + 4}{(s - 1)^2 + 1} + \frac{3}{s + 2}$$

The inverse Laplace transform of this expression is a linear combination of

$$\mathbf{e}^t \sin(t),\ \mathbf{e}^t \cos(t) \quad \text{and} \quad \mathbf{e}^{(-2\,t)}$$

To determine the coefficients of $\mathbf{e}^t \sin(t)$ and $\mathbf{e}^t \cos(t)$, we rewrite the numerator of the first term

$$\frac{7\,s + 4}{(s - 1)^2 + 1} + \frac{3}{s + 2} = \frac{7\,(s - 1) + 11}{(s - 1)^2 + 1} + \frac{3}{s + 2}$$

This means that

$$f(t) = 7\,\mathbf{e}^t \cos(t) + 11\,\mathbf{e}^t \sin(t) + 3\,\mathbf{e}^{(-2\,t)}$$

We verify this result by using Maple's **invlaplace** routine.

```
> e5:=simplify(invlaplace(e4, s, t));
```

$$e5 := 7\,\mathbf{e}^t \cos(t) + 11\,\mathbf{e}^t \sin(t) + 3\,\mathbf{e}^{(-2\,t)}$$

```
>
```

5: Page 383

Code the differential equation and the initial conditions.

```
> deq:=diff(w(t), t$2)+w(t)=t^2+2;
  ic:=w(0)=1, D(w)(0)=-1;
```

$$deq := \left(\frac{d^2}{dt^2} w(t)\right) + w(t) = t^2 + 2$$

$$ic := w(0) = 1,\ D(w)(0) = -1$$

Take the Laplace transform of both sides of the differential equation.

```
> e1:=laplace(deq, t, s);
```

$$e1 := s^2\,\text{laplace}(w(t), t, s) - D(w)(0) - s\,w(0) + \text{laplace}(w(t), t, s) = \frac{2}{s^3} + \frac{2}{s}$$

Substitute the initial conditions.

```
> e2:=subs({ic}, e1);
```

$$e2 := s^2\,\text{laplace}(w(t), t, s) + 1 - s + \text{laplace}(w(t), t, s) = \frac{2}{s^3} + \frac{2}{s}$$

Solve for the Laplace transform of $w(t)$.

```
> e3:=solve(e2, laplace(w(t), t, s));
```

$$e3 := \frac{-s^3 + s^4 + 2 + 2\,s^2}{s^3\,(s^2 + 1)}$$

Perform the inverse Laplace transform.

```
> e4:=invlaplace(e3, s, t);
```

$$e4 := \cos(t) - \sin(t) + t^2$$

```
>
```

18: Page 283

Code the differential equation and the initial conditions.

```
> deq:=diff(y(t), t$2)-2*diff(y(t), t)-y(t)=exp(2*t)-exp(t);
  ic:=y(0)=1, D(y)(0)=3;
```

$$deq := \left(\frac{d^2}{dt^2}\,y(t)\right) - 2\left(\frac{d}{dt}\,y(t)\right) - y(t) = e^{(2\,t)} - e^{t}$$

$$ic := y(0) = 1,\, D(y)(0) = 3$$

Take the Laplace transform of both sides of the differential equation.

```
> e1:=laplace(deq, t, s);
```

$$e1 := s^2\,\text{laplace}(y(t), t, s) - D(y)(0) - s\,y(0) - 2\,s\,\text{laplace}(y(t), t, s) + 2\,y(0) - \text{laplace}(y(t), t, s) = \frac{1}{(s-2)\,(s-1)}$$

Substitute the initial conditions.

```
> e2:=subs({ic}, e1);
```

$$e2 := s^2\,\text{laplace}(y(t), t, s) - 1 - s - 2\,s\,\text{laplace}(y(t), t, s) - \text{laplace}(y(t), t, s) = \frac{1}{(s-2)\,(s-1)}$$

Solve for the Laplace transform of y(t).

```
> e3:=solve(e2, laplace(y(t), t, s));
```

$$e3 := \frac{-2\,s^2 - s + 3 + s^3}{s^4 - 5\,s^3 + 7\,s^2 - s - 2}$$

Factor the result

```
> e4:=factor(e3);
```

$$e4 := \frac{-2\,s^2 - s + 3 + s^3}{(s-2)\,(s-1)\,(s^2 - 2\,s - 1)}$$

21: Page 383

Code the differential equation and the initial conditions.

```
> deq:=diff(y(t), t$2)-2*diff(y(t), t)+y(t)=cos(t)-sin(t);
  ic:=y(0)=1, D(y)(0)=3;
```

$$deq := \left(\frac{d^2}{dt^2}\,y(t)\right) - 2\left(\frac{d}{dt}\,y(t)\right) + y(t) = \cos(t) - \sin(t)$$

$$ic := y(0) = 1,\, D(y)(0) = 3$$

Take the Laplace transform of both sides of the differential equation.

```
> e1:=laplace(deq, t, s);
```

$$e1 := s^2\,\text{laplace}(y(t), t, s) - D(y)(0) - s\,y(0) - 2\,s\,\text{laplace}(y(t), t, s) + 2\,y(0) + \text{laplace}(y(t), t, s) = \frac{s-1}{s^2+1}$$

Substitute the initial conditions.

```
> e2:=subs({ic}, e1);
```

$$e2 := s^2\,\text{laplace}(y(t), t, s) - 1 - s - 2\,s\,\text{laplace}(y(t), t, s) + \text{laplace}(y(t), t, s) = \frac{s-1}{s^2+1}$$

Solve for the Laplace transform of y(t).

```
> e3:=solve(e2, laplace(y(t), t, s));
```

$$e3 := \frac{s\,(s + s^2 + 2)}{s^4 + 2\,s^2 - 2\,s^3 - 2\,s + 1}$$

Factor the result.

```
> e4:=factor(e3);
```

$$e4 := \frac{s\,(s + s^2 + 2)}{(s^2 + 1)\,(s - 1)^2}$$

```
>
```